
Your inventor box

Like a cook laying out ingredients, we check everything
before starting!

✓ 1 Arduino Uno board already programmed by TOUMA

✓ 1 HC-SR04 sensor: the robot’s “eyes”

✓ 1 L298N module: the “motor boss”

✓ 2 yellow motors + 2 wheels + 1 caster wheel

✓ 1 chassis, wires, 1 battery box

⚠ Always with an adult · batteries unplugged while wiring · touma.world

The wiring

One colour = one job. Red for +, black for ground (GND).

Part Arduino / L298N

HC-SR04 VCC 5V

HC-SR04 GND GND

HC-SR04 TRIG D9

HC-SR04 ECHO D10

L298N ENA / ENB D5 / D6

L298N IN1 / IN2 D2 / D3

L298N IN3 / IN4 D4 / D7

Left motor OUT1 / OUT2

Right motor OUT3 / OUT4

L298N GND Arduino GND (common ground)

⚠ Always with an adult · batteries unplugged while wiring · touma.world

First start

Robot on a book, wheels in the air: like an upside-down
bike to test the pedals.

1 Put the robot on a book, wheels free.

2 With the adult, plug in the batteries.

3 Wait 2 seconds: do the wheels spin forward?

4 One wheel spins backwards? Unplug the batteries and swap its two
wires.

5 Put your hand in front of the eyes: the wheels change direction.

⚠ Always with an adult · batteries unplugged while wiring · touma.world

Your robot’s code

This code is already in your board. You can copy it into
Arduino IDE to upload it again.

// TOUMA · Junior Inventeur · obstacle-avoiding robot
// TOUMA robot: drives forward if the path is clear (>= 20 cm), otherwise stops and turns right.
const int ENA = 5, ENB = 6, IN1 = 2, IN2 = 3, IN3 = 4, IN4 = 7;
const int TRIG = 9, ECHO = 10;

long distanceCm() {
  digitalWrite(TRIG, LOW); delayMicroseconds(2);
  digitalWrite(TRIG, HIGH); delayMicroseconds(10);
  digitalWrite(TRIG, LOW);
  long d = pulseIn(ECHO, HIGH, 30000);   // wait for the echo
  return d == 0 ? 400 : d / 58;          // nothing seen = far
}
void moteurs(int a, int b, int c, int d, int vitesse) {
  digitalWrite(IN1, a); digitalWrite(IN2, b);
  digitalWrite(IN3, c); digitalWrite(IN4, d);
  analogWrite(ENA, vitesse); analogWrite(ENB, vitesse);
}
void avancer()       { moteurs(HIGH, LOW, HIGH, LOW, 110); }
void tournerDroite() { moteurs(HIGH, LOW, LOW, HIGH, 110); }
void arreter()       { moteurs(LOW, LOW, LOW, LOW, 0); }

void setup() {
  int sorties[] = {ENA, ENB, IN1, IN2, IN3, IN4, TRIG};
  for (int i = 0; i < 7; i++) pinMode(sorties[i], OUTPUT);
  pinMode(ECHO, INPUT);
  arreter();
  delay(2000);                            // 2 s to put the robot down
}
void loop() {
  if (distanceCm() >= 20) {
    avancer();
  } else {
    arreter(); delay(300);
    tournerDroite(); delay(400);
  }
  delay(60);
}

⚠ Always with an adult · batteries unplugged while wiring · touma.world
Next, invent your own circuits in Toumi's Workshop.


